
ＰＣＧ センベーダー （Ver.181105K） 

 

・ 初期設定パート ： セーブスロット０に保存してください。 

1 'SVD181105K+PCG -INIT- 

10 CLV:CLP:CLS 

20 LET[0],#91E,#901,120,160,200,250:X=15:L=11 

31 POKE#814,#C,#27,#2F,#39,#F,#6,#9,#10 

32 POKE#81C,#30,#E4,#F4,#9C,#F0,#60,#90,#8 

33 POKE#824,#C,#7,#F,#39,#2F,#26,#9,#4 

34 POKE#82C,#30,#E0,#F0,#9C,#F4,#64,#90,#20 

35 POKE#710,#0,#10,#38,#BA,#FE,#FE,#BA,#0 

36 COPY#700,#814,16 

40 C=0:E=0:F=1:I=1:N=23:T=0:U=8:W=8:P=0:V=0 

50 FORA=LTOL-6STEP-2:LC1,A:FORB=1TO8:?"Ǡǡ ";:NEXT:NEXT:LC0,0:?"SCO

RE:";S; 

60 LRUNFILE()+1 

 

・ メインパート ： セーブスロット１に保存してください。 

1 'SVD181105K+PCG -MAIN- 

100 T=(T+1)%420 

110 IF!(T%2)X=X-(BTN(28)*(X>3))+(BTN(29)*(X<28)):LCX-1,23:?" Ǣ "; 

120 IF!(T%W)GSB300 

130 IFP=0ANDBTN(32)P=1:M=X:BEEP3:GSB200ELSEIFPGSB200 

140 IFEWAIT60:D=D+1:L=11+D:LRUNFILE()-1,40 

150 W=8-C/4 

160 GOTO100 

200 IF(T%W)LCM,N:?" "; 

210 N=N-1:IF!NP=0:N=23:RTN 

220 IFSCR(M,N)<=32LCM,N:?"|";:RTN 

230 LCM-1,N:?"***";:BEEP8:C=C+1:S=S+1:P=0:LC6,0:?S;:WAIT5:BEEP4:LC

M-1,N:?"   ";:N=23:E=C=32 

240 RTN 

300 IFPLCM,N:?" "; 

310 COPY#700,#814+I%2*16,16 

320 BEEP[I+2],5:I=(I+1)%4:B=6:FORA=LTOL-6*FSTEP-2 

330 COPY[V]+A*32,[V]+V*2-1+A*32,30*(V*2-1):[B]=SCR(!V*29+1,A):IF[B

]=32[B]=0 

340 B=B+1:NEXT 

350 IF[6]+[7]+[8]+[9]GSB400 

360 RTN 

400 B=1:FORA=0TO7:IFSCR(!V*8+A*3+V,L)=#E0THENB=0 

410 NEXT:IFBL=L-2:U=U-2:GOTO400 

420 COPY#91F+(L+1)*32,#91F+L*32,-32*U 

430 L=L+1:V=!V:IFC=31F=0:LET[7],0,0,0 

440 IFL=23LCX,23:?"Ǵ";:BEEP80,6:WAIT6:BEEP20:GOTO500 

450 RTN 

500 LC12,11:?"GAME OVER"; 

510 IFINKEY()=10LRUNFILE()-1:ELSEGOTO510 

 

・ 実行は必ずセーブスロット0のプログラムから行ってください（LOAD0してからRUN、又はLRUN0で実行）。 

 

※ プログラムはリロケータブル仕様になってます。暫定的にセーブスロット0と1と指定しましたが、初期設定パートを番号の小さいスロッ

ト、メインパートをその次の番号のスロットに保存というルールが守られていれば動作します。 

 

 

※ By nana_hachimaru@Twitter 


